Dynamics of a rocket with gravity $-GMm/r^2$

```
Clear["Global`*"]
```

Numerical values used for graphs and comparisons

*** Data for Lunar Modules (LM) of Apollo missions, and the Moon surface,
https://en.wikipedia.org/wiki/Ascent_propulsion_system ,
https://en.wikipedia.org/wiki/Apollo_Lunar_Module ,
https://nssdc.gsfc.nasa.gov/nmc/spacecraft/display.action?id=1972-031C ,
https://www.hq.nasa.gov/office/pao/History/SP-4205/ch6-5.html ,

m_dry = 2150 kg LM dry mass od the ascent stage,
m_p = 2350 kg propellant mass, m0 = mf+mp,
t_burn = 465 s burn time,
I_sp = 311 s   specific impulse,

F_thrust = ca.15.5 kN   (to be verified below),
deltav = 2200 m/s (to be verified below),

*** Moon and astronomy data
gm = G M_moon = 6.674E-11 Nm2/kg2 * 7.346E22 kg,
y0= R_moon = 1.74E6 m,
g_moon = 1.623 m/s2

```
params = {mp → 2350, mdry → 2150, tburn → 465, isp → 311, v0 → 0,
  gm → (6.674 × 10^-11 7.346 × 10^22), g0 → 9.81, rmoon → 1.74 × 10^6, gmoon → 1.623}
```

```
{mp → 2350, mdry → 2150, tburn → 465, isp → 311, v0 → 0,
 gm → 4.90272 × 10^12, g0 → 9.81, rmoon → 1.74 × 10^6, gmoon → 1.623}
```

```
(* m0  = m(t=0)  = *) m0 = mdry + mp;
m0 /. params
```

```
4500
```

```
ve = isp g0 ;
ve /. params
```

```
3050.91
```

```
(* r = dm/dt = *) r = mp/tburn;
r /. params // N
```

```
5.05376
```

```
(* fthrust should be equal about 15.5kN, fthrust=ve*R=  *) fthrust = ve r;
fthrust /. params
```

```
15418.6
```

```
(* Δv = ve Log[mf/m0]  should be about 2200 m/s *)
deltav = ve Log[m0/mdry]; deltav /. params
```

```
2253.43
```

Mass vs time, m (t), with r=dm/dt=mass flow rate (exhausted mass

per unit of time; r is assumed constant)

```
m[t_] := m0 - r t
```

```
Plot[m[t] /. params, {t, 0, tburn /. params},
 PlotLegends → {"m(t)"}, AxesLabel → {"t", "m(t)"}]
```



Rocket moving along a straight line OY towards infinity, no other masses yet (only for testing DSolve)

$$\text{sol0} = \text{DSolve}\left[\left\{y''[t] == \frac{\text{fthrust}}{m[t]},\ y[0] == 0,\ y'[0] == 0\right\}, y, t\right]$$

$$\left\{\left\{y \to \text{Function}\left[\{t\}, \frac{1}{\text{mp}} (\text{g0 isp mp t} + \text{g0 isp mp t Log}[\text{mdry tburn} + \text{mp tburn}] - \text{g0 isp mdry tburn Log}[\text{mdry tburn} + \text{mp tburn}] - \text{g0 isp mp tburn Log}[\text{mdry tburn} + \text{mp tburn}] - \text{g0 isp mp t Log}[-\text{mp t} + \text{mdry tburn} + \text{mp tburn}] + \text{g0 isp mdry tburn Log}[-\text{mp t} + \text{mdry tburn} + \text{mp tburn}] + \text{g0 isp mp tburn Log}[-\text{mp t} + \text{mdry tburn} + \text{mp tburn}])\right]\right\}\right\}$$

```
y'[t] /. sol0 // FullSimplify
```

{g0 isp (Log[(mdry + mp) tburn] - Log[-mp t + (mdry + mp) tburn])}

```
y'[t] /. sol0 /. params
```

$$\left\{\frac{1}{2350}\left(1.11516\times10^{8} - \frac{1.50025\times10^{13}}{2\,092\,500 - 2350\,t} + \frac{1.68487\times10^{10}\,t}{2\,092\,500 - 2350\,t} - 7.16964\times10^{6}\,\text{Log}[2\,092\,500 - 2350\,t]\right)\right\}$$

```
Plot[{y'[t] /. sol0 /. params, deltav /. params },
 {t, 0, tburn /. params}, PlotStyle -> {Thick, Dashed},
 PlotLegends → {"v(t) zero gravity", "Δv from rocket equation after t=t_burn"}]
```



```
y[t] /. sol0 // FullSimplify
```

$$\left\{\frac{1}{\text{mp}}\,\text{g0 isp}\,\left(\text{mp t} + (\text{mp t} - (\text{mdry} + \text{mp})\ \text{tburn})\ (\text{Log}[(\text{mdry} + \text{mp})\ \text{tburn}] - \text{Log}[-\text{mp t} + (\text{mdry} + \text{mp})\ \text{tburn}])\right)\right\}$$

```
Plot[{y[t] /. sol0 /. params}, {t, 0, tburn /. params},
 PlotStyle -> {Thick}, PlotLegends → {"y(t)  zero gravity"}]
```



Equations for g=const

$$\text{yc[t\_] := } -\frac{\text{gmoon } t^2}{2} + \text{t v0} + \frac{\text{fthrust}\left(t + \left(-\frac{\text{m0}}{r} + t\right)\text{Log}\left[\frac{\text{m0}}{\text{m0} - r\,t}\right]\right)}{r}$$

```
yc[t]
```

$$-\frac{\text{gmoon } t^2}{2} + \text{t v0} + \text{g0 isp}\left(t + \left(t - \frac{(\text{mdry} + \text{mp})\ \text{tburn}}{\text{mp}}\right)\text{Log}\left[\frac{\text{mdry} + \text{mp}}{\text{mdry} + \text{mp} - \frac{\text{mp t}}{\text{tburn}}}\right]\right)$$

Equations for g=const and m=m0=const

In[171]:= (* y0=0, v0=0, $y(t) = \frac{1}{2}\ a\ t^2$, a = Fnet/m0 *)

$$\text{ym0[t\_] := } \frac{1}{2}\ \frac{(\text{fthrust} - \text{m0 gmoon})}{\text{m0}}\ t^2$$

Rocket moving along a straight line OY from the surface of a planet/moon (y=0) towards infinity

(*

$a_y(t)$ =

dv_y/dt $a_y(t)=F_{net,y}$ / m(t)   where $F_{net,y}=F_{thrust}-G\ M\ m(t)/y^2(t)$ = $v_e R$ - $G\ M\ m(t)/y^2(t)$

and we assume that y=0 at the centre of a planet/moon of mass M

and we need to solve

$$\frac{d^2y(t)}{dt^2} = \frac{R\ v_e}{m(t)} - \frac{G\ M}{y^2(t)}$$

where $m(t) = m_0 - R\ t$, and $v_e=F_{thrust}/R$

*)

```
(* analytical solution of the differential eq. apparently not possible: *)
DSolve[{y''[t] == fthrust/m[t] - gm/(y[t])^2, y[0] == y0, y'[0] == 0}, y, t]
```

```
DSolve[{y''[t] == (g0 isp mp)/((mdry + mp - (mp t)/tburn) tburn) - gm/y[t]^2, y[0] == y0, y'[0] == 0}, y, t]
```

```
sol1 = NDSolve[{(y''[t] == fthrust/m[t] - gm/(y[t])^2) /. params, y[0] == rmoon /. params, y'[0] == 0},
  y, {t, 0, tburn /. params}]
```

```
{{y → InterpolatingFunction[{{0., 465.}}, <>]}}
```

In[174]:=
```
Plot[{y'[t] /. sol1, (y'[t]) /. sol0 /. params, (yc'[t]) /. params , (ym0'[t]) /. params },
 {t, 0, tburn /. params}, PlotStyle -> {Thick, Dashed, Dashed, Dashed}, PlotLegends →
  {"v(t)", "v(t) zero gravity", "v(t) g_moon=const", "v(t) g_moon=const, m=const"}]
```

Out[174]=



In[176]:=
```
Plot[{y[t] /. sol1, (y[t] + rmoon) /. sol0 /. params,
  (yc[t] + rmoon) /. params , (ym0[t] + rmoon) /. params}, {t, 0, tburn /. params},
 PlotStyle -> {Thick, Dashed, Dashed, Dashed}, PlotLegends →
  {"y(t)", "y(t) zero gravity", "y(t) g_moon = const", "y(t) g_moon=const, m=const"}]
```

Out[176]=




```
Table[{t, y[t]} /. sol1[[1]], {t, 0, 5}]
```

$\{\{0,\ 1.74\times10^{6}\},\ \{1,\ 1.74\times10^{6}\},\ \{2,\ 1.74\times10^{6}\},\ \{3,\ 1.74001\times10^{6}\},\ \{4,\ 1.74001\times10^{6}\},\ \{5,\ 1.74002\times10^{6}\}\}$

```
Export["v.txt", Table[{t, y'[t] /. sol1[[1]]}, {t, 0, tburn /. params}], "Table"]
Export["y.txt", Table[{t, y[t] /. sol1[[1]]}, {t, 0, tburn /. params}], "Table"]
```

```
v.txt
```

```
y.txt
```

```
Export["v-const-g.txt", Table[{t, (yc'[t]) /. params}, {t, 0, tburn /. params}], "Table"]
Export["y-const-g.txt",
 Table[{t, (yc[t] + rmoon) /. params}, {t, 0, tburn /. params}], "Table"]
```

```
v-const-g.txt
```

```
y-const-g.txt
```